unittest
{
	assert( [ "2", "0 0", "1 0" ].parse.expand.solve == "0.500000" );
	assert( [ "3", "0 0", "0 1", "1 0" ].parse.expand.solve == "0.707107" );
	assert( [ "10", "10 9", "5 9", "2 0", "0 0", "2 7", "3 3", "2 5", "10 0", "3 7", "1 9" ].parse.expand.solve == "6.726812" );
}



import std.conv;
import std.format;
import std.math;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	auto xys = new long[ 2 ][ n ];
	foreach( ref xy; xys )
	{
		input.popFront;
		xy = input.front.split.to!( long[] );
	}
	return tuple( xys );
}

auto solve( long[ 2 ][] xys )
{
	auto l = 0.0L;
	auto r = 1000.0L;
	foreach( i; 0 .. 50 )
	{
		auto c = ( l + r ) / 2;
		auto ok = true;
		
		// 全ての円が重なっているエリアが存在するか判定
		LOOP: foreach( ai, axy; xys )
		{
			foreach( bxy; xys[ ai + 1 .. $ ] )
			{
				ok = ( hypot( axy[ 0 ] - bxy[ 0 ], axy[ 1 ] - bxy[ 1 ] ) <= c * 2 );
				if( !ok ) break LOOP;
			}
		}
		
		if( ok ) r = c;
		else     l = c;
	}
	return format( "%0.6f", l );
}
